
/*************************************** 
  PHC 6080/6081 - ODS Basics          
****************************************/ 
 
Options nodate nonumber; 
 
/*  The dataset contains permanently stored formats,  
    you must submit the following code or use the  
    option NOFMTERR in the options statement above  
    to be able to use the WHAS500 SAS dataset       */ 
 
proc format; 
 value YesNoFmt 1='Yes' 
                0='No'; 
 value Sex01Ft  0='Male' 
                1='Female'; 
 value yr       1='1997' 
                2='1999' 
                3='2001'; 
 value ord      0='First' 
                1='Recurrent'; 
 value type     0='Non Q-wave' 
                1='Q-wave'; 
 value censA    0='Alive' 
                1='Dead'; 
run; 
 
 
/******************* 
 Check Dataset  
*******************/ 
 
proc contents data=temp.whas500; 
run; 
 
  



/* Although ODS was available in earlier versions of SAS, many new  
   features were added with the launch of version 9.2. This article 
   outlines some of the basic concepts of ODS and provides a few  
   examples of their use. There may be some changes in versions 9.3 
   and beyond 
 
     ODS GRAPHICS:       enchances output of standard SAS procedures by  
                        providing relevant high quality graphs - Automatic 
       in SAS 9.4+ - no need to request 

ODS LAYOUT:   Can be used to group output - I usually use for 
      graphs 
    ODS PDF/RTF(/HTML): sends output of standard SAS procedures to a  
                        document in the requested format 
        * Options available vary based upon the type of file created 
        * I find RTF useful if I want to modify tables and rearrange a  
            document or create a new document from the results 
        * I find PDF useful if I want to send output in a nice format  
            or take screen shots of nice looking output 
 ODS <doctype> TEXT: Adds text to document, formatting possible 
    ODS TRACE:          prints information in the log file detailing  
                        the internal names of each object in the  
                        output (these names can also be found in the  
                        SAS online documentation) 
    ODS SELECT/EXCLUDE: allows selection or exclusion of portions of  
                        the output 
    ODS OUTPUT:         Creates and stores a dataset from an object*/ 
 
 
/* Starting with 9.3 applicable procedures produce ODS graphics  
   by default if the setting is enabled in  
   TOOLS - OPTIONS - PREFERENCES in the RESULTS tab */ 
 
 
/* You can add text to the basic HTML Output using ODS HTML TEXT */ 
 
ods html text = "Test Text in HTML"; 
 
 
  



/********************************************** 
  Basic Code we will use for this discussion  
**********************************************/ 
 
proc glm data=temp.whas500; 
 model diasbp = sysbp; 
run; 
quit; 
 
/* ODS GRAPHICS  
 The ODS GRAPHICS statement can be used to change options.  
 ODS plot options can be requested in many procedures - here unpack and all.  
*/ 
 
ods graphics / height=3in width=4in; 
proc glm data=temp.whas500 plots(unpack)=all;  
 model diasbp = sysbp; 
run; 
quit; 
ods graphics / reset=all; /* reset graphs sizes to default */ 
 
/* I have had issues with getting layout to work but I am sure 
   it is my lack of understanding - what I can do is useful. */ 
 
ods graphics / height=2.25in width=3.25in; 
ods layout gridded columns=2 width=7in column_gutter=.15in ; 
 
ods region; /* column 1 region */ 
proc glm data=temp.whas500 plots(unpack)=all;  
 model diasbp = sysbp; 
 ods select ResidualHistogram QQPlot fitplot; 
run; 
quit; 
 
ods region; /* column 2 region */ 
proc glm data=temp.whas500 plots(unpack)=all;  
 model diasbp = sysbp; 
 ods select ResidualByPredicted ResidualPlots; 
run; 
quit; 
 
ods layout end; 
ods graphics / reset=all;  
 
  



 
/********************************************** 
 ODS RTF - send SAS output to an RTF file  
 We saved as ODS_Basics.rtf.    
 Notice that Layout does not work in RTF   
**********************************************/ 
 
ods rtf; 
 
proc glm data=temp.whas500 plots=none; 
 model diasbp = sysbp; 
run; 
quit; 
 
ods graphics / height=2.25in width=3.25in; 
ods layout gridded columns=2 width=7in column_gutter=.15in ; 
ods region;  
 
proc glm data=temp.whas500 plots(unpack)=all;  
 model diasbp = sysbp; 
 ods select ResidualHistogram QQPlot fitplot; 
run; 
quit; 
 
ods region;  
 
proc glm data=temp.whas500 plots(unpack)=all;  
 model diasbp = sysbp; 
 ods select ResidualByPredicted ResidualPlots; 
run; 
quit; 
 
ods layout end; 
ods graphics / reset=all;  
 
ods rtf close; 
  



 
/*  When SAS is done proccessing the request, a FILE DOWNLOAD window  
    will (should) pop up.  
    You can request numerous options, my favorites for RTF are: 
        FILE =      specify where the file is to be saved, if you do not  
                    specify a location, you will lose your file unless  
                    you save it from either the pop-up window or from  
                    inside the editing program once you open the file 
        BODYTITLE   puts titles in the body of the document instead of  
                    inside headers/footers 
        STYLE =     changes the look of the file 
        STARTPAGE = changes the way SAS determine page breaks in the document 
 
    The RTF file SAS creats can be viewed in WORD or WORDPAD. RTF files  
    can have very large file sizes. If you save the RTF file as a word  
    document, the file size is considerably reduced although the document 
    is less portable. 
*/ 
 
 
/*  If you get an error when trying the code above, you may 
    need to specify the file location directly.  You will still  
    get the error, but the file will be saved in the location 
    you specify (this is a SAS communication issue of some sort) 
*/ 
 
  



 
/********************************************** 
 Using file reference in ODS RTF 
**********************************************/ 
 
ods rtf file = "H:\_ODS\ODS_Basics2.rtf" style=printer bodytitle; 
 
Ods escapechar='^'; /* Tells SAS what character you want to use before commands */ 
/* the ^S is a style command and the ^4n is a line skipping command */ 
 
Ods rtf text="^4n  
  Adding Text (and space) to RTF file"; 
 
proc glm data=temp.whas500 ; 
 model diasbp = sysbp; 
run; 
quit; 
 
ods graphics / height=2.25in width=3.25in; 
ods layout gridded columns=2 width=7in column_gutter=.15in ; 
ods region;  
 
proc glm data=temp.whas500 plots(unpack)=all;  
 model diasbp = sysbp; 
 ods select ResidualHistogram QQPlot fitplot; 
run; 
quit; 
 
ods region;  
 
proc glm data=temp.whas500 plots(unpack)=all;  
 model diasbp = sysbp; 
 ods select ResidualByPredicted ResidualPlots; 
run; 
quit; 
 
ods layout end; 
ods graphics / reset=all;  
 
ods rtf close; 
 
  



 
/********************************************** 
 ODS PDF - send SAS output to a pdf file 
 Saved as ODS_Basics.pdf from file opened in SAS 
**********************************************/ 
 
/*  PDF files are created similarly although some options are different  
    for pdf files. You may notice that SAS automatically opens the pdf  
    file in the Results Viewer window. Similarly, additional options  
    can be requested. Below I have used NOTOC which supresses the  
    creation of the table of contents pane on the left, which I usually 
    do not find useful. If you get an error, try specifying the file name 
    directly.  The error will still exist (it is a firewall/SAS communication 
    issue of some sort) but your file will be saved where you requested 
*/ 
 
ods pdf  style=banker notoc ; 
 
proc glm data=temp.whas500 ; 
 model diasbp = sysbp; 
run; 
quit; 
 
ods pdf startpage = now; 
 
ods graphics / height=2.25in width=3.25in; 
ods layout gridded columns=2 width=7in column_gutter=.15in ; 
ods region;  
 
proc glm data=temp.whas500 plots(unpack)=all;  
 model diasbp = sysbp; 
 ods select ResidualHistogram QQPlot fitplot; 
run; 
quit; 
 
ods region;  
 
proc glm data=temp.whas500 plots(unpack)=all;  
 model diasbp = sysbp; 
 ods select ResidualByPredicted ResidualPlots; 
run; 
quit; 
 
ods layout end; 
ods graphics / reset=all;  
 
ods pdf close; 
  



 
/********************************************** 
 Using File reference in ODS PDF  
**********************************************/ 
 
ods pdf file = "H:\_ODS\ODS_Basics2.pdf" style=printer notoc; 
 
Ods escapechar='^';  
Ods PDF text="^4n"; 
Ods PDF text="^S={font=('Times Roman',12pt,Bold)  foreground=blue  
   background=orange just=c} Adding Text (and space) to PDF file"; 
 
proc glm data=temp.whas500; 
 model diasbp = sysbp; 
run; 
quit; 
 
ods pdf startpage = now; 
 
ods graphics / height=2.25in width=3.25in; 
ods layout gridded columns=2 width=7in column_gutter=.15in ; 
ods region;  
 
proc glm data=temp.whas500 plots(unpack)=all;  
 model diasbp = sysbp; 
 ods select ResidualHistogram QQPlot fitplot; 
run; 
quit; 
 
ods region;  
 
proc glm data=temp.whas500 plots(unpack)=all;  
 model diasbp = sysbp; 
 ods select ResidualByPredicted ResidualPlots; 
run; 
quit; 
 
ods layout end; 
ods graphics / reset=all;  
 
ods pdf close; 
 
  



 
/************************************************************************* 
  Each of the components of the output can be individually referenced  
  through the ODS system. The names are listed in the SAS documentation 
  but I usually find it easier to use the commands ODS TRACE ON and  
  ODS TRACE OFF to identify the exact results seen in the current output. 
  The names appear in the log file as the results are added to the output. 
  View the log file after running the code below.  
*************************************************************************/ 
 
 
/*********************** 
 ODS TRACE  
***********************/ 
 
ods trace on; 
proc glm data=temp.whas500; 
 model diasbp = sysbp; 
run; 
quit; 
ods trace off; 
 
 
/**************************** 
 ODS SELECT and EXCLUDE 
****************************/ 
 
ods pdf file="H:\_ODS\ODS_Basics3.pdf" notoc startpage=never; 
 
proc glm data=temp.whas500; 
 model diasbp = sysbp; 
 ods select parameterestimates fitplot; 
run; 
quit; 
 
ods pdf close; 
 
 
ods pdf file="H:\_ODS\ODS_Basics4.pdf" notoc startpage=no; 
 
ods graphics / height=3in width=4in; 
proc glm data=temp.whas500; 
 model diasbp = sysbp; 
 ods exclude nobs fitplot; 
run; 
quit; 
  



 
 
/***************************************** 
 Graphs of the variables individually  
*****************************************/ 
 
ods pdf startpage=now; 
 
ods graphics / height=2.75in width=3.5in; 
 
ods layout gridded columns=2 width=7in column_gutter=.15in ; 
ods region; /* column 1 region */ 
 
proc sgplot data=temp.whas500; 
 histogram diasbp; 
 density diasbp / type=normal; 
 density diasbp / type=kernel; 
run; 
 
ods region; /* column 2 region */ 
 
proc sgplot data=temp.whas500; 
 histogram sysbp; 
 density sysbp / type=normal; 
 density sysbp / type=kernel; 
run; 
 
ods region; /* column 1 region */ 
 
proc sgplot data=temp.whas500; 
 vbox diasbp; 
run; 
 
ods region; /* column 2 region */ 
 
proc sgplot data=temp.whas500; 
 vbox sysbp; 
run; 
 
ods region; /* column 1 region */ 
 
proc univariate data=temp.whas500 noprint; 
 qqplot diasbp; 
run; 
 
ods region; /* column 2 region */ 
 
proc univariate data=temp.whas500 noprint; 
 qqplot sysbp; 
run; 
 
ods layout end; 
ods graphics / reset=all;  
ods pdf close; 
 
  



 
/************ 
 ODS OUTPUT  
************/ 
 
/*  Any table can be output to a SAS dataset. These datasets can be used  
    for a variety of purposes. For the moment, consider that you will  
    run the same simple linear regression model for each gender and,  
    you want a table containing the gender, the r-squared value, the  
    estimated slope, its standard error and p-value. Using ODS output  
    and some reasonably basic data manipulation, we can accomplish this. 
 
    First, note that the two tables we will need are called FitStatistics  
    and ParameterEstimates. We will begin by exporting each of these 
    to their own dataset using ODS Output. In addition, the code prints 
    the two datasets to a pdf file. Normally, I would simply print these 
    to the screen to determine how to edit them further. The ODS PDF  
    contains a few more options including the TEXT= option which places  
    text in the document (which can be formatted). 
*/ 
 
proc sort data=temp.whas500; 
 by gender; 
run; 
 
proc glm data=temp.whas500; 
 by gender; 
 model diasbp = sysbp; 
 ods output ParameterEstimates=Parms FitStatistics=Fit; 
run; 
quit; 
 
 
/* The goal is to create a different presentations of the 
   Information in this output */ 
 
  



 
/******************* 
 View Datasets  
*******************/ 
 
ods pdf file="H:\_ODS\ODS_BasicsODSOUTPUT.pdf" notoc startpage=no; 
 
Ods escapechar='^';  
Ods PDF text="^3n"; 
Ods PDF text="^S={font=('Times Roman',12pt,Bold) just=c} Print of data=parms"; 
Ods PDF text="^1n"; 
 
proc print data=parms; 
run; 
 
title; /* clear title */ 
 
Ods PDF text="^3n"; 
Ods PDF text="^S={font=('Times Roman',12pt,Bold) just=c} Print of data=fit"; 
Ods PDF text="^1n"; 
 
proc print data=fit; 
run; 
 
title; 
 
 
/*  Use the printed datasets (or PROC CONTENTS) to determine how  
    to clean the datasets  
*/ 
 
  



 
/******************* 
 Clean Datasets  
*******************/ 
 
/*  The following SAS code will clean up the datasets by removing  
    rows and columns that are not needed and combine the two datasets 
    into one with a fairly simple merge. There are a number of neat  
    tricks in the code. Among them: 
        Retain statement to reorder variables 
        If-then statement 
        Rename statement 
        Merge statement 
*/ 
 
data parms; 
 retain Gender Estimate StdErr ProbT; /* using retain to reorder variables */ 
 set parms; 
 if Parameter="Intercept" then delete; /* IF works on observations */ 
 drop Parameter Dependent tValue; /* drop and keep on variables */ 
 * keep Gender Estimate StdErr ProbT; 
run; 
 
Ods PDF text="^3n"; 
Ods PDF text="^S={font=('Times Roman',12pt,Bold) just=c} Second Print of 
data=parms"; 
Ods PDF text="^1n"; 
 
proc print data=parms; 
run; 
 
data fit; 
 set fit; 
 keep gender RSquare RootMSE; 
run; 
 
Ods PDF text="^3n"; 
Ods PDF text="^S={font=('Times Roman',12pt,Bold) just=c} Second Print of 
data=fit"; 
Ods PDF text="^1n"; 
 
proc print data=fit; 
run; 
 
  



 
/*********************************** 
 Merge Datasets into Final Table  
***********************************/ 
 
data out; 
 retain Gender Rsquare Estimate StdErr ProbT; 
 merge parms fit; /* merge datasets - in simple situations adds variables 
(set statement adds observations) */ 
 by gender; /* all datasets in merge statement must be sorted by gender - 
ours are by default in same order here */ 
run; 
 
/* View Final Table */ 
Ods PDF text="^3n"; 
Ods PDF text="^S={font=('Times Roman',12pt,Bold) just=c} Print of Result of Merge, 
data=out"; 
Ods PDF text="^1n"; 
 
proc print data=out noobs; 
run; 
 
ods pdf close; 
 
 
/*************************** 
 Reset your HTML session  
***************************/ 
 
ods html close; 
ods html;  

 


